# Maintainer: Filipe Laíns (FFY00) <lains@archlinux.org>

_pkgname=pytest-regressions
pkgname=python-$_pkgname
pkgver=2.4.2
pkgrel=2
pkgdesc='Pytest plugin for regression testing'
arch=('any')
url='https://github.com/ESSS/pytest-regressions'
license=('MIT')
depends=('python-pytest' 'python-pytest-datadir' 'python-yaml')
makedepends=('python-build' 'python-installer' 'python-setuptools-scm' 'python-wheel')
source=("$pkgname-$pkgver.tar.gz::$url/archive/$pkgver.tar.gz")
sha512sums=('4014434b78bedb8b6a42d6a5156243dc0db8c9f78e0741cca32776e9768f04b0be0af5d8e6524025825b3e44efc394499c70cdc323393b9d56bebd5a88609da0')

build() {
  cd $_pkgname-$pkgver

  SETUPTOOLS_SCM_PRETEND_VERSION=$pkgver python -m build -nw
}

package() {
  cd $_pkgname-$pkgver

  python -m installer -d "$pkgdir" dist/*.whl

  install -Dm 644 LICENSE "$pkgdir"/usr/share/licenses/$pkgname/LICENSE
}

# vim:set ts=2 sw=2 et:
